Math 262a, Fall 1999, Glenn Tesler
Homework 5

```
> read `hsum.mpl`;
```

Copyright 1998 Wolfram Koepf, Konrad-Zuse-Zentrum Berlin

Koepf 6.7(e)

First, see if any of the earlier ways we learned to do this are applicable.

```
> Fe := k * binomial(n,k) * binomial(s,k);
  rhse := s*binomial(n+s-1,n-1);
```

$$Fe := k \operatorname{binomial}(n, k) \operatorname{binomial}(s, k)$$

$$rhse := s \operatorname{binomial}(n + s - 1, n - 1)$$

```
> gosper(Fe,k);
Error, (in gosper) no hypergeometric term antidifference exists
```

Creative telescoping

```
> sumrecursion(Fe,k=1..n,f(n));
```

$$-n \operatorname{f}(n+1) + (s+n) \operatorname{f}(n) = 0$$

```
> sumrecursion(Fe,k,f(n));
```

$$-n \operatorname{f}(n+1) + (s+n) \operatorname{f}(n) = 0$$

```
> rec_e := rsolve({",f(1)=A},f(n));
```

$$rec\_e := \frac{\Gamma(s+n)\, A}{\Gamma(n)\, \Gamma(s+1)}$$

```
> subs(A=eval(subs(k=1,n=1,Fe)), rec_e);
```

$$\frac{\Gamma(s+n)\, s}{\Gamma(n)\, \Gamma(s+1)}$$

```
> convert(",binomial);
```

$$s \operatorname{binomial}(s+n-1, s)$$

Sister Celine's algorithm

```
> fasenmyer(Fe,k,f(n),1);
Error, (in kfreerec) no kfree recurrence equation of order (, 1, 1, ) exists
> fasenmyer(Fe,k,f(n),2);
```

$$(n+1) \operatorname{f}(n+2) - (n+1+s) \operatorname{f}(n+1) = 0$$

It's a shift of the same recurrence found by CT, so the solution will be the same.

WZ method

```
> F := Fe / rhse;
```

$$F := \frac{k\,\mathrm{binomial}(n, k)\,\mathrm{binomial}(s, k)}{s\,\mathrm{binomial}(n+s-1, n-1)}$$

```
> R := WZcertificate(F,k,n);
```

$$R := \frac{(k-1)\,k}{(-n+k-1)\,(n+s)}$$

```
> G := F*R;
```

$$G := \frac{k^2\,\mathrm{binomial}(n, k)\,\mathrm{binomial}(s, k)\,(k-1)}{s\,\mathrm{binomial}(n+s-1, n-1)\,(-n+k-1)\,(n+s)}$$

Verify the **WZ equation** for this identity:

```
> WZeq := subs(n=n+1,F) - F  =  subs(k=k+1,G)-G;
```

$$WZeq := \frac{k\,\mathrm{binomial}(n+1, k)\,\mathrm{binomial}(s, k)}{s\,\mathrm{binomial}(s+n, n)} - \frac{k\,\mathrm{binomial}(n, k)\,\mathrm{binomial}(s, k)}{s\,\mathrm{binomial}(s+n-1, n-1)} =$$
$$\frac{(k+1)^2\,\mathrm{binomial}(n, k+1)\,\mathrm{binomial}(s, k+1)\,k}{s\,\mathrm{binomial}(s+n-1, n-1)\,(k-n)\,(s+n)}$$
$$- \frac{k^2\,\mathrm{binomial}(n, k)\,\mathrm{binomial}(s, k)\,(k-1)}{s\,\mathrm{binomial}(s+n-1, n-1)\,(-n-1+k)\,(s+n)}$$

```
> simpcomb(WZeq);
```

$$\frac{\Gamma(s)\,\Gamma(n+1)\,(-n\,s-s+k\,s+k\,n)\,\Gamma(s+1)\,\Gamma(n)}{\Gamma(s+1+n)\,\Gamma(s+1-k)\,\Gamma(k)^2\,\Gamma(n+2-k)\,k} =$$
$$\frac{\Gamma(s)\,\Gamma(n+1)\,(-n\,s-s+k\,s+k\,n)\,\Gamma(s+1)\,\Gamma(n)}{\Gamma(s+1+n)\,\Gamma(s+1-k)\,\Gamma(k)^2\,\Gamma(n+2-k)\,k}$$

```
> evalb(");
```

$$true$$

Or with purely rational arithmetic:

```
> simpcomb(lhs(WZeq)/F) = simpcomb(rhs(WZeq)/F);
```

$$-\frac{-n\,s-s+k\,s+k\,n}{(-n-1+k)\,(s+n)} = -\frac{-n\,s-s+k\,s+k\,n}{(-n-1+k)\,(s+n)}$$

```
> evalb(");
```

$$true$$

Now sum up the equation over k=1..n. These are natural bounds, so we may extend them.

Let f(n) = sum(F(n,k), k=1..infinity)

The left side has sum

```
> Sum('F'(n+1,k)-'F'(n,k),k=1..infinity) =
    f(n+1)-f(n);
```

$$\sum_{k=1}^{\infty} (\mathrm{F}(n+1,k) - \mathrm{F}(n,k)) = \mathrm{f}(n+1) - \mathrm{f}(n)$$

and the right side has sum

```
> Sum('G'(n,k+1)-'G'(n,k),k=1..infinity) =
    'G'(n,infinity) - 'G'(n,1);
```

$$\sum_{k=1}^{\infty} (\mathrm{G}(n,k+1) - \mathrm{G}(n,k)) = \mathrm{G}(n,\infty) - \mathrm{G}(n,1)$$

```
> limit(G,k=infinity);
```

$$\lim_{k\to\infty} \frac{k^2 \,\mathrm{binomial}(n,k)\,\mathrm{binomial}(s,k)\,(k-1)}{s\,\mathrm{binomial}(s+n-1,n-1)\,(-n-1+k)\,(s+n)}$$

On integers, F has finite k-support for each n, so G does too, so this limit should be 0, even though maple doesn't recognize that.

```
> limit(G,k=1);
```

$$0$$

Thus, for n=1,2,3,...,

```
> f(n+1)-f(n)=0;
```

$$\mathrm{f}(n+1) - \mathrm{f}(n) = 0$$

which proves

```
> Sum(F,k=1..infinity)=constant;
```

$$\sum_{k=1}^{\infty} \frac{k\,\mathrm{binomial}(n,k)\,\mathrm{binomial}(s,k)}{s\,\mathrm{binomial}(s+n-1,n-1)} = constant$$

Evaluate the constant using f(1) = F(1,1):

```
> subs(n=1,k=1,F); eval(");
```

$$\frac{\mathrm{binomial}(1,1)\,\mathrm{binomial}(s,1)}{s\,\mathrm{binomial}(s,0)}$$

$$1$$

So we have proved

```
> Sum(F,k=1..infinity)=1;
```

$$\sum_{k=1}^{\infty} \frac{k\,\mathrm{binomial}(n,k)\,\mathrm{binomial}(s,k)}{s\,\mathrm{binomial}(s+n-1,n-1)} = 1$$

Restricting the summation range to the support and rearranging gives, for n=1,2,3,...,

```
> Sum(Fe,k=1..n) = rhse;
```

$$\sum_{k=1}^{n} k\,\mathrm{binomial}(n,k)\,\mathrm{binomial}(s,k) = s\,\mathrm{binomial}(n+s-1,n-1)$$

```
>
```

The companion identity is obtained by summing the WZ equation over n instead of k. Let

```
> g(k) = Sum(G,n=1..infinity);
```

$$g(k) = \sum_{n=1}^{\infty} \frac{k^2 \operatorname{binomial}(n, k) \operatorname{binomial}(s, k) (k-1)}{s \operatorname{binomial}(s+n-1, n-1) (-n-1+k) (s+n)}$$

Summing the left side of the WZ equation over n=1,2,... gives

```
> Sum('F'(n+1,k)-'F'(n,k),n=1..infinity) =
    'F'(infinity,k)-'F'(1,k);
```

$$\sum_{n=1}^{\infty} (\mathrm{F}(n+1, k) - \mathrm{F}(n, k)) = \mathrm{F}(\infty, k) - \mathrm{F}(1, k)$$

When n=1 we have

```
> 'F'(1,k)=subs(n=1,F);
```

$$\mathrm{F}(1, k) = \frac{k \operatorname{binomial}(1, k) \operatorname{binomial}(s, k)}{s \operatorname{binomial}(s, 0)}$$

so

```
> 'F'(1,k)=piecewise(k=1,1,0);
```

$$\mathrm{F}(1, k) = \begin{cases} 1 & k = 1 \\ 0 & \mathit{otherwise} \end{cases}$$

As n -> infinity,

```
> 'F'(infinity,k)=limit(F,n=infinity);
```

$$\mathrm{F}(\infty, k) = \lim_{n \to \infty} \frac{k \operatorname{binomial}(n, k) \operatorname{binomial}(s, k)}{s \operatorname{binomial}(n+s-1, n-1)}$$

It won't do it automatically, let's do it ourselves.

```
> simpcomb(F);
```

$$\frac{\Gamma(n)\,\Gamma(s+1)\,\Gamma(n+1)\,\Gamma(s)}{k\,\Gamma(n+1-k)\,\Gamma(s+1-k)\,\Gamma(n+s)\,\Gamma(k)^2}$$

```
> F_n := select(has,",n);  F_no_n := ""/F_n;
```

$$F\_n := \frac{\Gamma(n)\,\Gamma(n+1)}{\Gamma(n+1-k)\,\Gamma(n+s)}$$

$$F\_no\_n := \frac{\Gamma(s+1)\,\Gamma(s)}{k\,\Gamma(s+1-k)\,\Gamma(k)^2}$$

As n goes to infinity, Gamma(n+A)/Gamma(n+B) ~ n^(A-B) so the as n->infinity, F(n,k) is asymptotically

```
> F_inf := n^(1-(1-k+s)) * F_no_n;
```

$$F\_inf := \frac{n^{(-s+k)}\,\Gamma(s+1)\,\Gamma(s)}{k\,\Gamma(s+1-k)\,\Gamma(k)^2}$$

which is

```
> 'F'(infinity,k)=piecewise(k>s,infinity,
  k=s,simpcomb(subs(k=s,F_inf)), k<s, 0);
```

$$\mathrm{F}(\infty, k) = \begin{cases} \infty & s < k \\ 1 & k = s \\ 0 & k < s \end{cases}$$

```
> k,s;
```

$$k, s$$

Combining all this, summing the left side of the WZ equation over n=1,2,... gives

```
> sumWZlhs := 'piecewise(s<k,infinity,
              's=k and k<>1', 1,
              'k<s and k<>1', 0,
              's=1 and k=1',0,
              '1<s and k=1', -1)':
  Sum('F'(n+1,k)-'F'(n,k),n=1..infinity) = sumWZlhs;
```

$$\sum_{n=1}^{\infty} (\mathrm{F}(n+1, k) - \mathrm{F}(n, k)) = \begin{cases} \infty & s < k \\ 1 & s = k \textbf{ and } k \neq 1 \\ 0 & k < s \textbf{ and } k \neq 1 \\ 0 & s = 1 \textbf{ and } k = 1 \\ -1 & 1 < s \textbf{ and } k = 1 \end{cases}$$

The sum is divergent for s<k, so restrict to k<=s.

Summing the right side of the WZ equation over n=1,2,... gives

```
> Sum('G'(n,k+1)-'G'(n,k),n=1..infinity) =
    g(k+1)-g(k);
```

$$\sum_{n=1}^{\infty} (\mathrm{G}(n, k+1) - \mathrm{G}(n, k)) = \mathrm{g}(k+1) - \mathrm{g}(k)$$

Combining the sum of the left side and the sum of the right side gives g(k+1)-g(k) = the 5 cases listed just above.

Now for each s, solve the resulting recurrence for g(k).

(A) For s>1, s>=k, we have

g(s+1)-g(s)=1; g(s)-g(s-1)=g(s-1)-g(s-2)=...=g(3)-g(2)=0; g(2)-g(1)=-1; g(1)-g(0)=g(0)-g(-1)=...=0

so given g(s+1), then g(k)=g(s+1)-1 for k=2,3,...,s; g(k)=g(s+1) for k=s+1 or 1,0,-1,-2,...

(B) For s=1: g(k)=g(s+1) for k<=s+1.

(C) For k<=s<1: g(k)=g(s+1)-1 for k<=s

Now find an initial value. All of these are expressed in terms of g(s+1).

```
> simpcomb(subs(k=s+1,G));
```

$$0$$

so its sum is g(s+1)=0. Then the three cases become

(A) For s>1: g(k)=-1 for k=2,3,...,s; g(k)=0 for k=s+1 or 1,0,-1,-2,...

(B) For s=1: g(k)=0 for k<=s+1

(C) For s<1: g(k)=-1 for k<=s, and g(s+1)=0.

Case (A) spelled out in full is: For integer s>1 and integer k<=s+1,

```
> Sum(G,n=1..infinity) = piecewise('k=s+1 or k<=1',0,'2<=k
  and k<=s',-1);
```

$$\sum_{n=1}^{\infty} \frac{k^2 \operatorname{binomial}(n,k)\operatorname{binomial}(s,k)(k-1)}{s\operatorname{binomial}(n+s-1,n-1)(-n+k-1)(n+s)} = \begin{cases} 0 & k=s+1 \textbf{ or } k\le 1 \\ -1 & 2\le k \textbf{ and } k\le s \end{cases}$$

or putting all n-free factors on the right side,

```
> rhs2 := s / (k^2 * binomial(s,k) * (k-1)):
  G2 := G*rhs2:
  Sum(G2,n=1..infinity) = piecewise('k=s+1 or
  k<=1',0,'2<=k and k<=s',-rhs2);
```

$$\sum_{n=1}^{\infty} \frac{\operatorname{binomial}(n,k)}{\operatorname{binomial}(n+s-1,n-1)(-n+k-1)(n+s)} = \begin{cases} 0 & k=s+1 \textbf{ or } k\le 1 \\ -\dfrac{s}{k^2\operatorname{binomial}(s,k)(k-1)} & 2\le k \textbf{ and } k\le s \end{cases}$$

Let's check it.

```
> gks := proc(k0,s0)
      global G,n;
      local G0;
      G0 := subs(k=k0,s=s0,G);
      sum(G0,n=1..infinity);
  end:
> linalg[matrix](6,6,(k,s)->gks(k,s));
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & -1 & -1 & -1 & -1 & -1 \\ 0 & 0 & -1 & -1 & -1 & -1 \\ 0 & 0 & 0 & -1 & -1 & -1 \\ 0 & 0 & 0 & 0 & -1 & -1 \\ 0 & 0 & 0 & 0 & 0 & -1 \end{bmatrix}$$

```
>
```

Koepf 8.5(a)

```
> gosper((3*k+2)/(k+2) * binomial(k,k/2),k);
Error, (in gosper) algorithm not applicable
> extended_gosper((3*k+2)/(k+2) * binomial(k,k/2),k);
```

$$\frac{\left(\frac{1}{2}k+1\right)(3k+2)\,\mathrm{binomial}\left(k,\frac{1}{2}k\right)}{\left(\frac{3}{2}k+1\right)(k+2)}+\frac{\left(\frac{1}{2}k+\frac{3}{2}\right)(3k+5)\,\mathrm{binomial}\left(k+1,\frac{1}{2}k+\frac{1}{2}\right)}{\left(\frac{3}{2}k+\frac{5}{2}\right)(k+3)}$$

8.5(b)

```
> extended_gosper((3*k+4)/(k+4)*binomial(k/2,k/4),k);
```

$$\frac{\left(\frac{1}{4}k+1\right)(3k+4)\,\mathrm{binomial}\left(\frac{1}{2}k,\frac{1}{4}k\right)}{\left(\frac{3}{4}k+1\right)(k+4)}$$

$$+\frac{\left(\frac{1}{4}k+\frac{5}{4}\right)(3k+7)\,\mathrm{binomial}\left(\frac{1}{2}k+\frac{1}{2},\frac{1}{4}k+\frac{1}{4}\right)}{\left(\frac{3}{4}k+\frac{7}{4}\right)(k+5)}$$

$$+\frac{\left(\frac{1}{4}k+\frac{3}{2}\right)(3k+10)\,\mathrm{binomial}\left(\frac{1}{2}k+1,\frac{1}{4}k+\frac{1}{2}\right)}{\left(\frac{3}{4}k+\frac{5}{2}\right)(k+6)}$$

$$+\frac{\left(\frac{1}{4}k+\frac{7}{4}\right)(3k+13)\,\mathrm{binomial}\left(\frac{1}{2}k+\frac{3}{2},\frac{1}{4}k+\frac{3}{4}\right)}{\left(\frac{3}{4}k+\frac{13}{4}\right)(k+7)}$$

```
>
```

Koepf 8.7(5.21)

```
> F0 :=
  hyperterm([3*a+1/2,3*a+1,-n],[6*a+1,-n/3+2*a+1],4/3,k);
  r := pochhammer(1/3,n/3)*pochhammer(2/3,n/3) /
  (pochhammer(1+2*a,n/3)*pochhammer(-2*a,n/3));
```

```
F := F0/r:
```

$$F0 := \frac{\mathrm{pochhammer}\left(3\,a+\frac{1}{2}, k\right)\mathrm{pochhammer}(3\,a+1, k)\,\mathrm{pochhammer}(-n, k)\left(\frac{4}{3}\right)^k}{\mathrm{pochhammer}(6\,a+1, k)\,\mathrm{pochhammer}\left(-\frac{1}{3}\,n+2\,a+1, k\right)k!}$$

$$r := \frac{\mathrm{pochhammer}\left(\frac{1}{3}, \frac{1}{3}\,n\right)\mathrm{pochhammer}\left(\frac{2}{3}, \frac{1}{3}\,n\right)}{\mathrm{pochhammer}\left(1+2\,a, \frac{1}{3}\,n\right)\mathrm{pochhammer}\left(-2\,a, \frac{1}{3}\,n\right)}$$

Compute the first few values of sum_k F0(n,k). Note that -n is an upper parameter, so the sum terminates at k=n unless a is chosen to make one of the denominator parameters also be a negative integer.

```
> sumF := nn -> sum(subs(n=nn,F),k=0..nn);
```

$$sumF := nn \to \sum_{k=0}^{nn} \mathrm{subs}(n=nn, F)$$

```
> sumF(1);
```

$$\frac{2}{3}\,\mathrm{pochhammer}\left(3\,a+\frac{1}{2}, 0\right)\mathrm{pochhammer}(3\,a+1, 0)\,\pi\sqrt{3}\,\mathrm{pochhammer}\left(1+2\,a, \frac{1}{3}\right)\mathrm{pochhammer}\left(-2\,a, \frac{1}{3}\right) \Big/ \left(\mathrm{pochhammer}(6\,a+1, 0)\,\mathrm{pochhammer}\left(\frac{2}{3}+2\,a, 0\right)\Gamma\left(\frac{2}{3}\right)\right)$$

$$-\frac{8}{9}\,\frac{\left(3\,a+\frac{1}{2}\right)(3\,a+1)\,\pi\sqrt{3}\,\mathrm{pochhammer}\left(1+2\,a, \frac{1}{3}\right)\mathrm{pochhammer}\left(-2\,a, \frac{1}{3}\right)}{(6\,a+1)\left(\frac{2}{3}+2\,a\right)\Gamma\left(\frac{2}{3}\right)}$$

```
> simplify(");
```

$$0$$

```
> 'simplify(sumF(nn))'$nn=1..6;
```

$$0, 0, 1, 0, 0, 1$$

```
>
```

Apply WZ method.

```
> WZcertificate(F,k,n);
Error, (in WZcertificate) extended WZ method fails
```

I looked at the code, it doesn't attempt to figure out if it should use m-fold hypergeometric functions; you have to specify m if you want it done.

```
> R := WZcertificate(F,k,n,3);
```

$$R := -\frac{(6\,a+k)\,(-n+6\,a+3\,k)\,k}{(-n-1+k)\,(-n-2+k)\,(-n-3+k)}$$

Verify it:

```
> G := R*F:
```

The WZ equation becomes   F(n+3,k) - F(n,k) = G(n,k+1) - G(n,k)

```
> simpcomb( (subs(n=n+3,F)-F)  -  (subs(k=k+1,G) - G));
```

$$0$$

Let

```
> f(n) = Sum('F(n,k)',k=0..infinity);
```

$$\mathrm{f}(n) = \sum_{k=0}^{\infty} \mathrm{F}(n,\,k)$$

The left side of the WZ equation sums to f(n+3)-f(n), and the right side sums to G(n,infinity)-G(n,0).
The k-support of G is finite for each n, since G(n,k+1)/G(n,k) is

```
> ratio(G,k);
```

$$2\,\frac{(6\,a+1+2\,k)\,(3\,a+1+k)\,(-n-3+k)}{k\,(-n+6\,a+3\,k)\,(6\,a+k)}$$

which vanishes at k=n+3 (unless a=n/6 - k/2 for an integer k>n+3, yielding a pole in this).
So G(n,infinity)=0.
Also, G(n,0) is

```
> subs(k=0,G);
```

$$0$$

Thus, the WZ equation summed over k yields f(n+3)-f(n)=0-0=0   for integers n>=0.
From the initial conditions above, we have

```
> f(n) = piecewise('n mod 3'=0,1,0);
```

$$\mathrm{f}(n) = \begin{cases} 1 & n \bmod 3 = 0 \\ 0 & \mathit{otherwise} \end{cases}$$

so for integers n>=0,

```
> Sum('F[0](n,k)/r(n)',k=0..infinity)=rhs(");
```

$$\sum_{k=0}^{\infty} \frac{F_0(n,\,k)}{\mathrm{r}(n)} = \begin{cases} 1 & n = 0 \\ 0 & \mathit{otherwise} \end{cases}$$

so

```
> Sum('F[0](n,k)',k=0..infinity)=piecewise('n mod
```

```
3'=0,'r(n)',0);
```

$$\sum_{k=0}^{\infty} F_0(n, k) = \begin{cases} r(n) & n \bmod 3 = 0 \\ 0 & \textit{otherwise} \end{cases}$$

and plugging everything in,

```
> Sum(F0,k=0..infinity)=piecewise('n mod 3'=0,r,0);
```

$$\sum_{k=0}^{\infty} \frac{\mathrm{pochhammer}\left(3\,a+\frac{1}{2}, k\right) \mathrm{pochhammer}(3\,a+1, k)\, \mathrm{pochhammer}(-n, k) \left(\frac{4}{3}\right)^k}{\mathrm{pochhammer}(6\,a+1, k)\, \mathrm{pochhammer}\left(-\frac{1}{3}\,n+2\,a+1, k\right) k!} =$$

$$\begin{cases} \dfrac{\mathrm{pochhammer}\left(\frac{1}{3}, \frac{1}{3}\,n\right) \mathrm{pochhammer}\left(\frac{2}{3}, \frac{1}{3}\,n\right)}{\mathrm{pochhammer}\left(1+2\,a, \frac{1}{3}\,n\right) \mathrm{pochhammer}\left(-2\,a, \frac{1}{3}\,n\right)} & n \bmod 3 = 0 \\ 0 & \textit{otherwise} \end{cases}$$

```
>
>
```

Koepf 11.7

Recall that

```
> erf(x) = 2/sqrt(Pi) * Int(exp(-t^2), t=0..x);
```

$$\mathrm{erf}(x) = 2\,\frac{\int_0^x \mathbf{e}^{(-t^2)}\,dt}{\sqrt{\pi}}$$

Use Taylor series:

```
> 2/sqrt(Pi) * Int(Sum((-t^2)^k/k!,k=0..infinity),t=0..x);
```

$$2\,\frac{\int_0^x \sum_{k=0}^{\infty} \frac{(-t^2)^k}{k!}\,dt}{\sqrt{\pi}}$$

Doing it term by term gives

```
> 2/sqrt(Pi)*Sum((-1)^k * x^(2*k+1) / (k! *
  (2*k+1)),k=0..infinity);
>
```

$$2\,\frac{\displaystyle\sum_{k=0}^{\infty} \frac{(-1)^k\, x^{(2\,k+1)}}{k!\,(2\,k+1)}}{\sqrt{\pi}}$$

Convert the sum to hypergeometric notation

```
> 2/sqrt(Pi) * Sumtohyper((-1)^k * x^(2*k+1) /
  (k!*(2*k+1)),k);
```

$$2\,\frac{x\ \mathrm{Hypergeom}\left(\left[\frac{1}{2}\right], \left[\frac{3}{2}\right], -x^2\right)}{\sqrt{\pi}}$$

```
>
>
```

reset its meaning, it’s used from scratch below.

```
> F := ’F’;
```

$$F := F$$

Problem 3.

The summand is

```
> Fxk := x^(2*k+1)/(2*k+1)!;
```

$$Fxk := \frac{x^{(2\,k+1)}}{(2\,k+1)!}$$

Clearly (Ek Dx^2 - 1) annihilates this. We can try to find a mixed recurrence/diffeq whose coefficients are k-free by brute force.:

```
> # celinexk(F,x,k,xmax,kmax)
  # celinexk(F,x,k,xmax,kmax,verbose,c)
  #  Apply Celine’s alg. to function F of a continuous
  variable x and a discrete variable k.
  #  Use derivatives (d/dx)^0,...,(d/dx)^xmax, and shifts
  (Ek)^0,...,(Ek)^kmax.
  # optional:
  #  verbose can be true or false, indicating whether to
  print intermediate results.
  #  c  can be x or k, to collect with respect to it.
  Default k.
  celinexk := proc(F,x,k,xmax,kmax)
      local oper, recdiffeq, dF,
            i,j, Dx, Ek,
            eqs,vars,sols,
            verbose,c,arg;
```

```
    verbose := false; c := k;
    for arg in args[6..nargs] do
       if type(arg,boolean) then verbose := arg else c
:= arg fi;
    od;
    Dx := cat(‘D‘,x); Ek := cat(‘E‘,k);
    oper := 0;
    recdiffeq := 0;
    for i from 0 to xmax do
    for j from 0 to kmax do
       oper := oper + a[i,j] * Dx^i * Ek^j;
       if i=0 then dF := F else dF := diff(F,x$i) fi;
       recdiffeq := recdiffeq +
a[i,j]*simpcomb(subs(k=k+j,dF)/F);
    od od;

    if verbose then print(‘trial equation‘,recdiffeq)
fi;

    recdiffeq := numer(recdiffeq);
    recdiffeq := collect(recdiffeq,c);
    if verbose then print(‘collected in powers of
‘,c,recdiffeq) fi;

    eqs := {coeffs(recdiffeq,c)};
    vars := {'('a[i,j]'$'i'=0..xmax)'$'j'=0..kmax};
    sols := solve(eqs,vars);
    if verbose then print(‘final recursion/diffeq
operator:‘) fi;
    if sols=NULL then RETURN(FAIL)
    else oper := subs(sols,oper); fi;
end:
```

```
> celinexk(Fxk,x,k,1,1);
```

$$0$$

```
> celinexk(Fxk,x,k,2,2);
```

$$a_{0,0} + a_{0,1}\,Ek - a_{0,0}\,Dx^2\,Ek - a_{0,1}\,Dx^2\,Ek^2$$

Same thing, calculations spelled out:

```
> celinexk(Fxk,x,k,2,2,true);
```

$$\textit{trial equation},\ a_{0,0} + \frac{1}{2}\,\frac{a_{0,1}\,x^2}{(k+1)\,(2\,k+3)} + \frac{1}{4}\,\frac{a_{0,2}\,x^4}{(k+1)\,(2\,k+3)\,(k+2)\,(2\,k+5)}$$

$$+\frac{a_{1,0}(2k+1)}{x}+\frac{1}{2}\frac{a_{1,1}x}{k+1}+\frac{1}{4}\frac{a_{1,2}x^3}{(k+1)(2k+3)(k+2)}+2\frac{a_{2,0}(2k+1)k}{x^2}+a_{2,1}+\frac{1}{2}\frac{a_{2,2}x^2}{(k+1)(2k+3)}$$

collected in powers of , k, $64\,a_{2,0}k^6+(480\,a_{2,0}+32\,a_{1,0}x)k^5$

$$+(1360\,a_{2,0}+16\,a_{2,1}x^2+240\,a_{1,0}x+16\,a_{0,0}x^2)k^4$$
$$+(680\,a_{1,0}x+112\,a_{0,0}x^2+112\,a_{2,1}x^2+1800\,a_{2,0}+8\,a_{1,1}x^3)k^3+$$
$$(284\,a_{0,0}x^2+4\,a_{0,1}x^4+900\,a_{1,0}x+48\,a_{1,1}x^3+284\,a_{2,1}x^2+1096\,a_{2,0}+4\,a_{2,2}x^4)$$
$$k^2+(18\,a_{2,2}x^4+18\,a_{0,1}x^4+2\,a_{1,2}x^5+548\,a_{1,0}x+308\,a_{0,0}x^2+94\,a_{1,1}x^3$$
$$+240\,a_{2,0}+308\,a_{2,1}x^2)k+a_{0,2}x^6+20\,a_{2,2}x^4+5\,a_{1,2}x^5+120\,a_{1,0}x+20\,a_{0,1}x^4$$
$$+120\,a_{0,0}x^2+120\,a_{2,1}x^2+60\,a_{1,1}x^3$$

final recursion/diffeq operator:

$$a_{0,0}+a_{0,1}Ek-a_{0,0}Dx^2Ek-a_{0,1}Dx^2Ek^2$$

```
> collect(",{a[0,0],a[0,1]},distributed);
```

$$(Ek-Dx^2Ek^2)\,a_{0,1}+(1-Dx^2Ek)\,a_{0,0}$$

The two operators are related by Ek-Dx^2 Ek^2 = Ek*(1-Dx^2 Ek), so just use the second one. Note it was also possible to see this by inspection.

The desired mixed recurrence/diffeq is (RDE)

```
> F(x,k) - diff(F(x,k+1),x$2) = 0;
```

$$\mathrm{F}(x,k)-D_{1,1}(F)(x,k+1)=0$$

Let

```
> f(x) = Sum(F(x,k),k=-infinity..infinity);
```

$$\mathrm{f}(x)=\sum_{k=-\infty}^{\infty}\mathrm{F}(x,k)$$

Sum (RDE) for k=-infinity..+infinity to get

```
> f(x) - diff(f(x),x$2)=0;
```

$$\mathrm{f}(x)-\left(\frac{\partial^2}{\partial x^2}\mathrm{f}(x)\right)=0$$

whose solution is

```
> dsolve(",f(x));
```

$$\mathrm{f}(x)=\_C1\,\mathbf{e}^{x}+\_C2\,\mathbf{e}^{(-x)}$$

The initial conditions f(0)=0, f'(0)=1 are easily checked, so use them:

```
> dsolve({"",f(0)=0,D(f)(0)=1},f(x));
```

$$f(x) = \frac{\frac{1}{2}(\mathbf{e}^x)^2 - \frac{1}{2}}{\mathbf{e}^x}$$

```
> simplify(");
```

$$f(x) = \sinh(x)$$

```
>
```

Koepf 12.1

```
> Fnt := t^n * exp(-t^2 - x/t);
  celinexk(Fnt,t,n,1,1,t);
```

$$Fnt := t^n\, \mathbf{e}^{\left(-t^2 - \frac{x}{t}\right)}$$

$$0$$

```
> celinexk(Fnt,t,n,2,2,t);
```

$$0$$

```
> celinexk(Fnt,t,n,3,3,t);
```

$$-a_{1,2}\, x + (-2\, a_{1,2} - a_{1,2}\, n)\, En + 2\, a_{1,2}\, En^3 - a_{2,2}\, x\, Dt + (-2\, a_{2,2} - a_{2,2}\, n)\, Dt\, En + a_{1,2}\, Dt\, En^2 + 2\, a_{2,2}\, Dt\, En^3 - a_{3,2}\, x\, Dt^2 + (-a_{3,2}\, n - 2\, a_{3,2})\, Dt^2\, En + a_{2,2}\, Dt^2\, En^2 + 2\, a_{3,2}\, Dt^2\, En^3 + a_{3,2}\, Dt^3\, En^2$$

```
> collect(",{a[1,2],a[2,2],a[3,2]});
```

$$((-n-2)\, En - x + Dt\, En^2 + 2\, En^3)\, a_{1,2} + ((-n-2)\, Dt\, En + Dt^2\, En^2 + 2\, Dt\, En^3 - x\, Dt)\, a_{2,2} + ((-n-2)\, Dt^2\, En + 2\, Dt^2\, En^3 - x\, Dt^2 + Dt^3\, En^2)\, a_{3,2}$$

```
> subs(a[2,2]=0,a[3,2]=0,a[1,2]=1,");
```

$$(-n-2)\, En - x + Dt\, En^2 + 2\, En^3$$

```
> rde := ":
```

So we have (-n-2) F(n+1,t) - x*F(n,t) + (d/dt) F(n+2,t) + 2 F(n+3,t) = 0.
(The x is suppressed from the parameter list of F.)
Integrate with respect to t. The terms are:
integral of (-n-2) F(n+1,t) dt: (-n-2) A(n+1,x)
integral of -x*F(n,t) dt: -x*A(n,x)
integral of (d/dt) F(n+2,t) dt = F(n+2,infinity)-F(n+2,0) = 0-0 = 0
integral of 2 F(n+3,t) dt = 2 A(n+3,x)

```
> -(n+2) * A(n+1,x) - x*A(n,x) + 0 + 2*A(n+3,x) = 0;
```

$$-(n+2)\,A(n+1,x)-x\,A(n,x)+2\,A(n+3,x)=0$$

There is also a command in Koepf's software that would have done this for us:

```
> intrecursion(Fnt,t,A(n,x));
```

$$-2\,A(n+3)+(n+2)\,A(n+1)+A(n)\,x=0$$

(It doesn't understand the extra parameter x.)

```
>
```

```
>
```

I couldn't get a diff eq using Sister Celine's algorithm.
Koepf's hsum package has a routine based on the continuous Gosper algorithm, and it produces

```
> intdiffeq(t^n*exp(-t^2-x/t),t,A(x));
```

$$x\left(\frac{\partial^3}{\partial x^3}A(x)\right)-(n-1)\left(\frac{\partial^2}{\partial x^2}A(x)\right)+2\,A(x)=0$$

```
>
```